
MAX3664EUA I/O Model

SPICE I/O Macromodels aid in understanding signal integrity issues in electronic systems.
Most of Maxim’s High Frequency/Fiber Communication ICs utilize input and output (I/O)
circuits with Current Mode Logic (CML), Positive Emitter Coupled Logic (PECL), and
Low Voltage Differential Signal (LVDS) formats to transfer data into and out of an IC.
These models are based on simplified circuit expressions that may include replacement of
active circuit elements with ideal controlled voltage and current sources. As such,
simulation with macromodels should be treated as ‘typical’ performance and
not relied upon as final proof-of-design. Use of macromodel descriptions is not a
substitute for worst-case design analysis, nor for testing real circuits over temperature,
supply, and other operating limits.

The output format is provided as ASCII text netlists suitable for generic SPICE. This
format is compatible with most versions of SPICE such as PSPICE and HSPICE.
Additional information is found in HFAN 6.1 Input/Output Models for Maxim Fiber
Components.

To extract the circuit netlists using the Adobe Acrobat Reader follow these instructions.
Select the "Text Select Tool" by clicking the left mouse button on this icon of the menu
bar (a capital T with a small dashed box to the lower right). Highlight the desired netlist
text with the cursor. Use the copy command from the edit menu to capture the selected
lines. Then paste the selected lines into a text file editor and save the file with an extension
compatible with the simulator.

 Figure 1. Output signal buffer for the MAX3664 TIA  including a simplified package model.

Notes:

The schematic on the previous page represents the output stage of the Maxim
MAX3664EUA Transimpedance amplifier. The input stage is not modeled since the
electrical model is not considered an adequate representation for the analysis of the input
optical signal.

The netlist is in SPICE 2g6 format. Since nodes in SPICE 2g6 can only be numbers, the
output signals are 2001 and 2002. Comments in the netlist identify the correspondence
between the signal names and the node numbers. The netlists are in SPICE 2g6 format and
are compatible with PSPICE and HSPICE. It has been simulated on a generic SPICE
simulator and PSPICE.

The Output Stage: The output stage is comprised  of the subcircuit driver OUT_DRV
and package model OUTPKG. If the user wishes a model for the die part only, then
removing the package components are easily done.

Text File Format: This model is shipped in “pdf” format. Models and netlists can be
copied to text format in the Acrobat Reader by holding the left mouse button on the “Text
Select Tool.” Then the user can “select” what netlist and/or subcircuit with the mouse.
Then use the copy command from the “edit” menu to capture the selected lines. These
lines can then be “pasted” into the user’s text file.



Circuit Netlist – Output Circuit

INPUT -  MAX3265 OUTPUT CIRCUIT
*
* THIS IS THE TYPICAL CML OUTPUT OF THE MAX3265
*
.OPTIONS ACCT NOMOD NOPAGE LIMPTS=10000 RELTOL=.001
.WIDTH OUT=80
.TEMP 35
* TYPICAL DIE TEMP = 25C + 2.2W*(26C/W) = 80C
.OP
.TRAN 5PS 8000PS
*
* CONVENTIONS VCC = 101, VEE = 102, + OUT = 2000, - OUT = 2002
*
VCC 101 0 DC 3.3
E90 90 0 2001 2002 1
E91 91 0 2002 2001 1
RLOAD1 2001 2002 100
CLOAD1 2001 101 0.20P
CLOAD2 2002 101 0.20P
CLOAD3 2001 2002 0.05P
XPK1 2001 2002 1001 1002 OUTPKG

.SUBCKT OUTPKG 2001 2002 1001 1002
CPKGP 2001 0 200F
CPKGN 2002 0 200F
RPKGP 2001 1003 100M
LPKGP 1003 1001 5N
RPKGN 2002 1004 100M
LPKGN 1004 1002 5N
.ENDS OUTPKG

XCIROUT 1001 1002 101 OUTDRV
*
.SUBCKT OUTDRV 1001 1002 101
VINP 2 0 PULSE (1.7 0.55 0.2N 0.30N 0.30N 1.3N 3.00N)
VINN 3 0 PULSE (0.55 1.7 0.2N 0.30N 0.30N 1.3N 3.00N)
*
RB1 2 22 1200
RB2 3 23 1200
*
XQOUTP 101 22 40 0 H14E04
XQOUPN 101 23 41 0 H14E04

ROUTP 40 1001 40
ROUTN 41 1002 40
*
IE1 1001 0 25M
IE2 1002 0 25M
*

XPAD1 1001 0 HPAD3
XPAD2 1002 0 HPAD3
XESD1P 1001 1 0 HDE113032
XESD1M 0 1001 0 HDE113032
XESD2P 1002 1 0 HDE113032
XESD2M 0 1002 0 HDE113032
*



.ENDS OUTDRV
*
*
** BEGINNING OF PROCESS LIB
*

.SUBCKT DESD 1 2 21
CP1EPI 1 4 8.743F
QD 5 4 1 5 QESD
*dd 1 4 dcb : area=count
*ds 5 4 dsub : area=count
RS 4 2 32.058 TC=2.813M,2.043U
RSUB 5 21 16.621K
CTRENCH 2 5 6.455F
.MODEL QESD PNP( IS=6.109E-019 NF=1.050 BF=800M BR=600U CJE=12.463F
+   VJE=640M MJE=330M CJC=5.346F VJC=790M MJC=460M )
.ENDS DESD

.SUBCKT PADESD100 2 3 4 5
CXP1 2 5 50F
XQ1 2 3 5 DE381011
XQ2 4 2 5 DE381011
.ENDS PADESD100

.SUBCKT DE381011 1 2 21
CP1EPI 1 4 132.715F
QD 5 4 1 5 QESD
RS 4 2 2.024 TC=2.615M,1.746U
RSUB 5 21 2.318K
CTRENCH 2 5 38.731F
.MODEL QESD PNP( IS=1.080E-017 NF=1.050 BF=800M BR=600U CJE=220.280F
+   VJE=640M MJE=330M CJC=75.512F VJC=790M MJC=460M )
.ENDS DE381011
*
*
.SUBCKT H11M052 1 2 3 21
CP1EPI 1 2 2.192F
CP1P2 12 3 6.473F
CTRENCH 1 20 7.654F
RBX 2 12 82.961 TC=2.214M
RCX 1 10 40.687 TC=2.961M,1.642U
RCI 10 11 2.141 TC=2.961M,1.642U
REX 13 3 7.711 TC=57.819U
RSUB 20 21 14.233K
QP 20 10 12 20 TXP OFF
QN 11 12 13 11 TX
*XREPORT1 0 REPORTERL1N13
*XREPORT2 0 REPORTERL1N8
.MODEL TX NPN( IS=1.151E-017 XTI=3 EG=1.140 BF=223.719 BR=20 XTB=450M
+   VAF=29 VAR=3.500 NF=1.010 NR=1.020 NE=1.650 NC=1.560 IKF=44.016M
+   IKR=806.400U ISE=5.444E-021 ISC=7.000E-030 RB=82.961 RBM=62.221
+   IRB=7M CJE=37.151F MJE=490M VJE=940M FC=990M CJC=7.359F MJC=470M
+   VJC=850M TF=3.814P TR=19N XTF=1 VTF=1K ITF=20.430M PTF=5 KF=1.500F
+   AF=1 )
.MODEL TXP PNP( IS=6.540E-019 CJE=7.359F MJE=470M VJE=850M CJC=7.985F
+   MJC=400M VJC=650M BF=10K BR=924.230U TF=1N FC=900M )
.ENDS H11M052
*
*
.SUBCKT H14E04 1 2 3 21



CP1EPI 1 2 12.185F
CP1P2 12 3 19.634F
CTRENCH 1 20 15.307F
RBX 2 12 20.554 TC=2.663M
RCX 1 10 23.509 TC=2.354M,979.573N
RCI 10 11 1.237 TC=2.354M,979.573N
REX 13 3 2.795 TC=123.150U
RSUB 20 21 6.195K
QP 20 10 12 20 TXP OFF
QN 11 12 13 11 TX
*XREPORT1 0 REPORTERL1N45
*XREPORT2 0 REPORTERL1N46
.MODEL TX NPN( IS=3.326E-017 XTI=3 EG=1.140 BF=232.533 BR=20 XTB=450M
+   VAF=29 VAR=3.500 NF=1.010 NR=1.020 NE=1.650 NC=1.560 IKF=126.546M
+   IKR=2.318M ISE=1.573E-020 ISC=2.022E-029 RB=20.554 RBM=15.416
+   IRB=20.223M CJE=107.672F MJE=490M VJE=940M FC=990M CJC=21.929F
+   MJC=470M VJC=850M TF=3.778P TR=19N XTF=1 VTF=1K ITF=59.621M PTF=5
+   KF=1.500F AF=1 )
.MODEL TXP PNP( IS=1.968E-018 CJE=21.929F MJE=470M VJE=850M CJC=21.582F
+   MJC=400M VJC=650M BF=10K BR=809.067U TF=1N FC=900M )
.ENDS H14E04
*
*
.SUBCKT HPAD3 1 3
CPAD 1 10 86.407F
REPI 10 20 149.204M TC=4.800M,5U
CTRENCH 21 20 79.795F
DS 21 20 DSUB
RS 3 21 369.115
*XREPORT1 0 REPORTERL1N94
.MODEL DSUB D( IS=98.719F CJO=555.750F M=400M VJ=650M )
.ENDS HPAD3
*
.SUBCKT HDE113032 1 2 21
CP1EPI 1 4 88.881F
QD 5 4 1 5 QESD
RS 4 2 2.531 TC=2.729M,1.896U
RSUB 5 21 2.936K
CTRENCH 2 5 22.961F
.MODEL QESD PNP( IS=1.181E-017 NF=1.050 BF=800M BR=600U CJE=155.018F
+   VJE=600M MJE=400M CJC=53.258F VJC=650M MJC=400M )
.ENDS HDE113032
*

*
.PRINT TRAN V(2001) V(2002)
*.PROBE
*
.END


